Limite et continuité d'une fonction à une variable

© Pierre Lantagne
Enseignant retraité du Collège de Maisonneuve

La première version de ce document est parue à l'automne 1998. Ce document actualisé renferme des éléments de programmation Maple qui ne sont pas destinés à l'apprentissage de Maple par l'élève. Ces éléments sont, par contre, utile à la présentation des concepts mathématiques de la limite et de la continuité.

Bonne lecture à tous !

* Ce document Maple est exécutable avec la version 2020.1

Initialisation

```
> restart;
> with(plots,setoptions):
  setoptions(size=[300,300],axesfont=[times,roman,8],color=navy ):
> with(Student[Calculus1]):
```

L'initialisation suivante permettra d'avoir plus de lisibilité des nombres décimaux en supprimamt les zéros non significatifs à la fin d'un nombre.

```
> interface(typesetting=extended); # Pour s'assurer le niveau de
  composition étendue
  Typesetting:-Settings(striptrailing=true);
```

$$extended$$

$$false \tag{1.1}$$

```
> Digits:=20;
```

$$Digits := 20 \tag{1.2}$$

Présentation inituitive de la notion de limite

Saississons la fonction f définie par $f(x) = \dfrac{x^3 - 4x^2 + 3x}{x - 3}$.

```
> f:=x->(x^3-4*x^2+3*x)/(x-3);
```

$$f := x \mapsto \frac{x^3 - 4 \cdot x^2 + 3 \cdot x}{x - 3} \tag{2.1}$$

Traçons le graphique de cette fonction .

```
> plot([x,f(x),x=-3..4]);
```



Sur le graphique ci-haut, il semble que la fonction f soit une fonction définie partout sur $[-3, 4]$. On n'observe aucun trou, aucune coupure.

Calculons f(3):

```
> f(3);
Error, (in f) numeric exception: division by zero
```

On pouvait s'y attendre puisque le domaine de cette fonction rationnelle est $\mathbb{R} \setminus \{3\}$. Analytiquement, examinons les points de la fonction f à l'aide des valeurs d'abcisses x voisines de la valeur 3. Pour commencer, construisons une suite de nombres voisins de 3 mais toujours inférieurs à la valeur 3 de telle manière que $|x-3| \rightarrow 0$. On dit que ce sont des voisins à gauche de 3:

```
> formule:=3-10^(-i);
  suite:=[seq(formule,i=1..8)];
```

$$formule := 3 - 10^{-i}$$

$$suite := \left[\frac{29}{10}, \frac{299}{100}, \frac{2999}{1000}, \frac{29999}{10000}, \frac{299999}{100000}, \frac{2999999}{1000000}, \frac{29999999}{10000000}, \frac{299999999}{100000000}\right] \quad \textbf{(2.2)}$$

Maple opère très bien l'arithmétique des nombres rationnels sous la forme de fractions. Mais pour notre propos, opérons les nombres rationnels sous leur écriture décimale.

```
> abscisses:=evalf(suite);
```

$$abscisses := [2.9, 2.99, 2.999, 2.9999, 2.99999, 2.999999, 2.9999999, 2.99999999] \quad \textbf{(2.3)}$$

Dans cette liste limitée, il est clair que $|x-3| \rightarrow 0$.
Obtenons maintenant les valeurs des images de ces voisins de 3.

```
> ordonnées:= map(f,abscisses);
```

$$ordonnées := [5.51, 5.9501, 5.995001, 5.99950001, 5.9999500001, 5.999995, 5.9999995, 5.99999995] \quad \textbf{(2.4)}$$

Voici donc la liste des points obtenus.

```
> points:=zip((x,y)->[x,y],abscisses,ordonnées);
```

$$points := [[2.9, 5.51], [2.99, 5.9501], [2.999, 5.995001], [2.9999, 5.99950001], [2.99999, 5.9999500001], [2.999999, 5.999995], [2.9999999, 5.9999995], [2.99999999, 5.99999995]] \quad \textbf{(2.5)}$$

On constate ceci: lorsque la valeur de x est de plus en plus voisine de 3, tout en étant inférieure à 3 où $|x-3|$

$\to 0$, la valeur des images de ces voisins est de plus en plus proche de 6. On utilise la notation mathématique suivante pour exprimer ce fait.

```
> Limit(f(x),x=3,left)=limit(f(x),x=3,left);
```

$$\lim_{x \to 3-} \frac{x^3 - 4x^2 + 3x}{x - 3} = 6 \tag{2.6}$$

REMARQUE: le signe moins ($-$) apparaissant à côté du nombre 3 signifie que les voisins de 3 sont inférieurs à 3 avec $|x-3| \to 0$. Ces voisins immédiats ne sont pas des nombres négatifs.

Dans le même ordre d'idée, mais cette fois-ci avec des valeurs de x voisines à droite de 3.

```
> formule:=3+10^(-i);
  suite:=[seq(formule,i=1..8)];
```

$$formule := 3 + 10^{-i}$$

$$suite := \left[\frac{31}{10}, \frac{301}{100}, \frac{3001}{1000}, \frac{30001}{10000}, \frac{300001}{100000}, \frac{3000001}{1000000}, \frac{30000001}{10000000}, \frac{300000001}{100000000}\right] \tag{2.7}$$

```
> abscisses:=evalf(suite);
```

$$abscisses := [3.1, 3.01, 3.001, 3.0001, 3.00001, 3.000001, 3.0000001, 3.00000001] \tag{2.8}$$

Obtenons maintenant les valeurs des images de ces voisins de 3.

```
> ordonnées:= map(f,abscisses);
```

$$ordonnées := [6.51, 6.0501, 6.005001, 6.00050001, 6.0000500001, 6.000005, 6.0000005, 6.00000005] \tag{2.9}$$

Voici donc la liste des points obtenus.

```
> points:=zip((x,y)->[x,y],abscisses,ordonnées);
```

$$points := [[3.1, 6.51], [3.01, 6.0501], [3.001, 6.005001], [3.0001, 6.00050001], [3.00001, 6.0000500001], [3.000001, 6.000005], [3.0000001, 6.0000005], [3.00000001, 6.00000005]] \tag{2.10}$$

On constate ceci: lorsque la valeur de x est de plus en plus voisine de 3, tout en étant supérieure à 3 où $|x-3| \to 0$, la valeur des images de ces voisins de 3 est de plus en plus proche de 6. On utilise la notation mathématique suivante pour exprimer ce fait;

```
> Limit(f(x),x=3,right)=limit(f(x),x=3,right);
```

$$\lim_{x \to 3+} \frac{x^3 - 4x^2 + 3x}{x - 3} = 6 \tag{2.11}$$

Lorsque des voisins d'un nombre a, incluant les voisins à gauche de a et à sa droite, ont leur image de plus en plus voisine du même nombre L, on utilise la notation de limite en omettant le signe + et le signe -.

```
> Limit(f(x),x=3)=limit(f(x),x=3);
```

$$\lim_{x \to 3} \frac{x^3 - 4x^2 + 3x}{x - 3} = 6 \tag{2.12}$$

En résumé, dans la macro-commande `limit`, si la « direction » n'est pas spécifiée et si l'évaluateur donne pour résultat un nombre réel, on peut affirmer donc que la limite existe et vaut ce nombre.

La limite d'une fonction f en $x = a$ existe si et seulement si pour tout voisinage de $x = a$, $\lim_{x \to a-} f(x) = L = \lim_{x \to a+} f(x)$ où $L \in \mathbb{R}$.

REMARQUE: Dans l'évaluation d'une limite en $x = a$ d'une fonction réelle à valeurs réelles, il n'y a que deux approches directionnelles vers x. Et lorsqu'une de ces limites "directionnelles" existe et vaut $L \in \mathbb{R}$, la définition même de la limite ne précise en rien de la manière dont les images de f s'pprochent de L.

Traitons maintenant un exemple de calcul de limite où la limite demandée en $x = a$ n'existe pas bien que les limites à gauche et à droite existent toutes deux.

Soit la fonction g définie par $g(x) = \frac{|x|}{x}$. Clairement, le domaine de la fonction g est $\mathbb{R} \setminus \{0\}$. Calculons d'abord la limite à gauche en $x = 0$.

```
> g:=x->abs(x)/x;
  Limit(g(x),x=0,left)=limit(g(x),x=0,left);
```

$$g := x \mapsto \frac{|x|}{x}$$

$$\lim_{x \to 0-} \frac{|x|}{x} = -1 \tag{2.13}$$

Calculons ensuite la limite à droite en $x = 0$.

```
> g:=x->abs(x)/x;
  Limit(g(x),x=0,right)=limit(g(x),x=0,right);
```

$$g := x \mapsto \frac{|x|}{x}$$

$$\lim_{x \to 0+} \frac{|x|}{x} = 1 \tag{2.14}$$

Puisque $\lim_{x \to 0-} g(x) \neq \lim_{x \to 0+} g(x)$, la limite $\lim_{x \to 0} g(x)$ n'existe pas.

Observons comment l'évaluateur traitera la requête du calcul de la limite en $x = 0$ sans spécifier la « direction ».

```
> Limit(g(x),x=0)=limit(g(x),x=0);
```

$$\lim_{x \to 0} \frac{|x|}{x} = \mathit{undefined} \tag{2.15}$$

L'évaluateur répond, à sa manière, que la limite demandée n'existe pas. Et nous savons pourquoi.

Traçons le graphique de la fonction g sur l'intervalle $[-5, 5]$.

```
> plot([x,g(x),x=-5..5],discont=true);
```



Voici un dernier exemple de calcul de limite avant que nous n'abordions le concept de la continuité en un point.

Soit la fonction g définie par $g(x) = \frac{\sqrt{1-x}+x}{1-x}$. Pour que $g(x)$ puisse être définie, il faut, d'une part, que le calcul de $1-x \geq 0$ et d'autre part, que $1-x \neq 0$. Il faut donc que $1-x>0$ Le domaine de la fonction g est donc $[-\infty, 1[$. Cela suffit pour conclure que la $\lim_{x\to 1} g(x)$ n'existe pas puisque la $\lim_{x\to 1+} g(x)$ n'existe pas. (Il n'y a aucun voisinage à droite de 1 pour approcher 1) Qu'en est-il de la limite à gauche $\lim_{x\to 1-} g(x)$?

```
> g:=x->(sqrt(1-x)+x)/(1-x);
  Limit(g(x),x=1,left)=limit(g(x),x=1,left);
```

$$g := x \mapsto \frac{\sqrt{1-x}+x}{1-x}$$

$$\lim_{x\to 1-} \frac{\sqrt{1-x}+x}{1-x} = \infty \qquad \textbf{(2.16)}$$

La limite $\lim_{x\to 1-} g(x)$ n'existe pas en dépit du signe égalité: la notation ∞ est un concept et non pas un nombre. J'aimais bien dire à mes étudiants que le symbole ∞ est une vue de l'esprit et si on débattait s'il s'agissait d'un nombre, je leur demandais quel est-il ? L'utilisation du signe égal est seulement pour nous informer que lorsque x prend des valeurs de plus en plus voisines de 1, tout en étant inférieures à 1 et tel que $|x-1| \to 0$, les images de ces voisines deviennent, quant à elles, de plus en plus grandes et ne se rapproche pas d'un certain nombre $L \in \mathbb{R}$.

Traçons donc le graphique de la fonction g sur l'intervalle $]-5, 1[$ "pour voir".

```
> plot([x,g(x),x=-5..1]);
```



Bref, pour $x<1$, $f(x) \to \infty$ lorsque $|x-1| \to 0$.

Plus loin dans ce document, nous nous préoccuperons de la manière dont $f(x) \to \infty$.

Définition de la continuité en $x = a$ d'une fonction f

La continuité est une notion importante et complexe en mathématique. C'est une notion se situant au coeur du développement du Calcul. On peut présenter intituivement la notion de la continuité d'une fonction de manière géométrique comme ceci: une fonction continue en $x = a$ est celle dont le graphique ne présente aucune interruption $x = a$, aucun trou $x = a$, aucun saut $x = a$, aucune asymptote verticale $x = a$.

Gardons tout de même à l'esprit que cette manière *géométrique* de caractériser la notion de la continuité est *imparfaite*. Cette présentation peu formelle est reprise plus rigoureusement de manière analytique comme suit:

Une fonction f est dite continue en $x = a$ si et seulement si $\lim_{x \to a} f(x) = f(a)$. La vérification de cette égalité comporte trois éléments afin de montrer que l'égalité est satisfaite:

1) Il faut que $f(a)$ soit définie, autement dit que $a \in dom(f)$
2) Il faut que $\lim_{x \to a} f(x)$ existe
3) Il faut que $\lim_{x \to a} f(x) = f(a)$

On a aussi la définition suivante: Une fonction f est dite continue en $x = a$ si et seulement si $\lim_{h \to 0} f(a+h) = f(a)$

Par définition donc, si l'égalité $\lim_{x \to a} f(x) = f(a)$ n'est pas vérifiée, la fonction f n'est pas continue en $x = a$.
On dit aussi de la fonction f qu'elle est discontinue en $x = a$.
Afin d'être en mesure de définir la continuité d'une fonction f sur un intervalle $I \subseteq dom(f)$, il est nécessaire de définir la continuité à gauche et la continuité à droite de f en $x = a$. Cela découle de l'existence des limites directionnelles de f en $x = a$.

Une fonction f est dite continue à gauche en $x = a$ si et seulement si $\lim_{x \to a-} f(x) = f(a)$
Une fonction f est dite continue à droite en $x = a$ si et seulement si $\lim_{x \to a+} f(x) = f(a)$

De ces deux définitions, on peut affirmer qu'une fonction f est continue en $x = a$ si et seulement si la fonction f est, à la fois, continue à gauche et à droite en $x = a$.

Une fonction f est continue sur un intervalle $I \subseteq dom(f)$ que si elle est continue $\forall\, x \in I$. Lorsqu'il s'agit de la continuité aux bornes de *I*, il s'agit de la continuité à gauche ou à droite, selon les cas où les extrémités de *I* sont incluses ou non dans *I*. C'est le cas si *I* est fermé ou semi-fermé: $I = [a, b]$, $I =]a, b]$, $I = [a, b[$.

La discontinuité d'une fonction f en $x = a$ peut être caractérisée de deux façons: discontinuité *essentielle* et discontinuité *non essentielle*.

– cette discontinuité en $x = a$ est essentielle si $\lim_{x \to a} f(x)$ n'existe pas

– cette discontinuité en $x = a$ est ***non essentielle*** si $\lim_{x \to a} f(x)$ existe

Traçons, de nouveau, la fonction de la section précédente définie pas $f(x) = \dfrac{x^3 - 4x^2 + 3x}{x - 3}$.

```
> plot([x,f(x),x=-3..4]);
```



Le tracé du graphe de cette fonction ne nous donne pas nécessairement des informations claires quant à la continuité de celle-ci. Il semble que cette fonction ne possède aucune discontinuité.

Calculons f(3).

```
> f(3);
Error, (in f) numeric exception: division by zero
```

Puisque $f(3)$ n'est pas définie, l'égalité $\lim_{x \to 3} f(x) = f(3)$ ne peut donc être vérifiée. La fonction f est donc discontinue en $x = 3$. Rappelons que la $\lim_{x \to 3} \frac{x^3 - 4x^2 + 3x}{x - 3}$ existe:

```
> Limit((x^3-4*x^2+3*x)/(x-3),x=3)=limit((x^3-4*x^2+3*x)/(x-3),x=3);
```

$$\lim_{x \to 3} \frac{x^3 - 4x^2 + 3x}{x - 3} = 6 \quad \mathbf{(3.1)}$$

La fonction f possède donc un discontinuité non essentielle en $x = 3$. Puisque f(3) n'est pas définie, la fonction f présente un trou, on parle alors d'une discontinuité par trou.
Si on définissait f(3) de telle manière que $\lim_{x \to 3} f(x) \neq f(3)$, cette discontinuité non essentielle serait qualifiée de discontinuité par déplacement.

```
> discont(f(x),x);
```

$$\{3\} \quad \mathbf{(3.2)}$$

La macro-commande Maple `discont` renvoie tous les points de discontinuité sur les réels. Cela inclut les points où la fonction va à plus ou moins l'infini. Donc, ce résultat ne nous informe pas sur la nature de la discontinuité: essentielle ou non essentielle.

Dans le cas d'une discontinuité par trou, il est possible de prolonger la définition de la fonction f en donnant à $f(3)$ la valeur de $\lim_{x \to 3} f(x)$ pour en faire une fonction continue en $x = 3$. Ainsi la fonction f nouvellement définie par

$$f(x) = \begin{cases} 6 & x = 3 \\ \dfrac{x^3 - 4x^2 + 3x}{x - 3} & x \neq 3 \end{cases}$$

est continue en $x = 3$. Nous venons de prolonger f par continuité en $x = 3$.

Rigoureusement, en procédant ainsi, nous faisons un abus de notation. En effet, la fonction f, que nous prolongeons par continuité en $x = 3$, n'est pas définie en ce réel. Lorsque nous la prolongeons par continuité, le domaine de définition de la fonction change et donc la fonction (mathématiquement parlant) change. Nous devrions changer de notation en toute rigueur.

En pratique, il y a très souvent intérêt à prolonger (lorsque cela est possible) une fonction par continuité car la fonction obtenue (étant continue) va satisfaire des théorèmes et propriétés facilitant l'étude mathématique.

Le graphique de la fonction $y = f(x) = \dfrac{x^3 - 4x^2 + 3x}{x - 3}$ ressemble beaucoup au graphique d'une fonction quadratique, c'est-à-dire à une parabole. En effet, observons le calcul de simplification suivant.

```
> f(x);
```

$$\frac{x^3 - 4x^2 + 3x}{x - 3} \quad \textbf{(3.3)}$$

```
> simplify(f(x));
```

$$(-1 + x)\, x \quad \textbf{(3.4)}$$

On constate donc que le calcul de $\dfrac{x^3 - 4x^2 + 3x}{x - 3}$ est égale au calcul de $(x - 1)\, x$ qu'à la condition évidemment que $x \neq 3$.

Exercice Maple, vérifions cela en posant, dans les deux expressions précédentes, $x = \sqrt{2}$.

```
> 'f(sqrt(2))'=f(sqrt(2));
```

$$f(\sqrt{2}) = \frac{5\sqrt{2} - 8}{\sqrt{2} - 3} \quad \textbf{(3.5)}$$

Rationnalisons $f(\sqrt{2})$.

```
> ``= radnormal(rhs((3.5)),'rationalized');
```

$$= 2 - \sqrt{2} \quad \textbf{(3.6)}$$

Et

```
> Eval((x-1)*x,x=sqrt(2))=eval((x-1)*x,x=sqrt(2));
  expand(rhs(%));
```

$$\left.(-1 + x)\, x\right|_{x = \sqrt{2}} = (\sqrt{2} - 1)\sqrt{2}$$

$$2 - \sqrt{2} \quad \textbf{(3.7)}$$

Traçons maintenant le graphique de la fonction "en exagérant" la discontinuité réductible en x=3.

```
> Trou:=plottools[disk]([3,6],0.15,color=white):
  Courbe:=plot([x,f(x),x=-3..4]):
  plots[display]([Courbe,Trou],scaling=constrained);
```



Remarque: Lors du tracé d'un graphique, Maple calcule les coordonnées d'un certain nombre de points selon un algorithme. Rappelons qu'ensuite, il relie ces points selon son mécanisme d'interpolation. Alors, pour des graphiques avec discontinuités réductibles, c'est seulement par analyse qu'il est possible de trouver les valeurs de discontinuité de la fonction.

Autres exemples de discontinuité essentielles (par saut, infinie et par oscillations)

Discontinuité par saut

Voici un premier exemple d'une fonction discontinue par saut. Traçons la fonction partie entière de x: $y = \lfloor x \rfloor$.

```
> f:=x->floor(x);
```

$$f := x \mapsto \lfloor x \rfloor \quad \textbf{(4.1.1)}$$

```
> plot([x,f(x),x=-3..5]);
```



Ce tracé n'est pas correct. Maple a tracé des droites presque verticales autour des points de discontinuité (c'est dû au mécanisme d'interpolation). L'option `discont=true` indiquera à Maple de faire attention à ce type de discontinuité et ainsi de tracer correctement le graphique.

```
> plot([x,f(x),x=-3..5],discont=true);
```



Cette fonction, dont le domaine est l'ensemble de tous les nombres réels, présente une infinité de points de discontinuité: il s'agit de tous les points de la fonction pour lesquels l'abscisse est un nombre entier. C'est la deuxième condition de la continuité qui n'est pas satisfaite.

La fonction partie entière fournit donc un exemple de fonction définie sur $\mathbb{R}$ et discontinue pour $\forall\, x \in \mathbb{Z}$ (et donc pas continue sur $\mathbb{R}$).

Voici un autre exemple d'une fonction discontinue par saut.

```
> f:=x->x-floor(x);
```

$$f := x \mapsto x - \lfloor x \rfloor \tag{4.1.2}$$

```
> plot([x,f(x),x=-1..3],view=[-1..3,0..1.5],discont=true);
```



Voici un dernier exemple obtenu à l'aide de la fonction *piecewise.* (fonction définie par morceaux).

```
> f:=t->piecewise(t<1,t^2,t>=1 and t<2,1,3);
  'f'(t)=simplify(f(t));
```

$$f := t \mapsto \begin{cases} t^2 & t < 1 \\ 1 & 1 \le t < 2 \\ 3 & \textit{otherwise} \end{cases}$$

$$f(t) = \begin{cases} t^2 & t < 1 \\ 1 & t < 2 \\ 3 & 2 \le t \end{cases} \tag{4.1.3}$$

Remarque: il est intéressant de constater la manière dont Maple a simplifié l'énoncé de cette fonction. L'ordre des énoncés conditionnels sur les valeurs de la variable est TRÈS important. À la première condition vraie, Maple retourne immédiatement une valeur.

Traçons le graphique de f.

```
> plot([t,f(t),t=-3..4],discont=true);
```



Cette fonction est continue partout sauf en $x = 2$. Car...

```
> f(2);
```

$$3 \quad \textbf{(4.1.4)}$$

```
> limit(f(t),t=2);
```

$$\mathit{undefined} \quad \textbf{(4.1.5)}$$

c'est la deuxième condition qui n'est pas satisfaite.

En effet, la limite n'existe pas car la limite à gauche n'est pas égale à la limite à droite.

```
> Limit('f'(t),t=2,left)=limit(f(t),t=2,left);
```

$$\lim_{t \to 2-} f(t) = 1 \quad \textbf{(4.1.6)}$$

```
> Limit('f'(t),t=2,right)=limit(f(t),t=2,right);
```

$$\lim_{t \to 2+} f(t) = 3 \quad \textbf{(4.1.7)}$$

Discontinuité infinie

Considérons la fonction $y = \dfrac{3}{(x+2)^2}$.

```
> g:=x->3/(x+2)^2:
  plot([x,g(x),x=-3..1]);
```



On a que dom f = $\mathbb{R} \setminus \{-2\}$ puisque $f(-2)$ n'est pas définie, la fonction est donc discontinue en $x = -2$. La première condition de la continuité n'est pas satisfaite.

On observe, par ailleurs, que pour les valeurs de x de plus en plus voisines de -2, tout en étant inférieures à -2, lorsque $|x-(-2)| \to 0$, leurs images ont des valeurs de plus en plus grandes: ces images ne deviennent pas de plus en plus voisines d'un certain nombre $L \in \mathbb{R}$. Donc la limite à gauche ***n'existe pas***.

Dans ce cas, même si la limite à gauche n'existe pas, il est permis d'utiliser cette notation de limite pour donner une information quant au comportement des images de ces valeurs de x voisines de -2:

```
> Limit('g'(x),x = -2,left)=limit(g(x),x = -2,left);
```

$$\lim_{x \to (-2)^-} g(x) = \infty \quad \textbf{(4.2.1)}$$

Il faut donc comprendre cet énoncé comme ceci: lorsque les valeurs de x sont de plus en plus voisines de -2, tout en étant inférieures à -2, lorsque $|x-(-2)| \to 0$ leurs images prennent des valeurs de plus en plus grandes.

De manière similaire, demandons à Maple de calculer la limite à droite:

```
> Limit('g'(x),x = -2,right)=limit(g(x),x = -2,right);
```

$$\lim_{x \to (-2)^+} g(x) = \infty \quad \textbf{(4.2.2)}$$

En résumé, puisque le comportement des images des voisins de -2, aussi bien les voisins de gauche que ceux de droite, est le même, on peut écrire:

```
> Limit('g'(x),x = -2)=limit(g(x),x = -2);
```

$$\lim_{x \to -2} g(x) = \infty \quad \textbf{(4.2.3)}$$

Dans les manuels, on a l'habitude, dans de tels cas de discontinuité, de tracer également une droite verticale qu'on appelle asymptote verticale pour mettre davantage l'emphase sur le comportement des images des valeurs voisines de -2. Ajoutons manuellemement le tracé de l'asymptote verticale d'équation $x = -2$.

```
> DroiteV:=plot([-2,y,y=-1..100],linestyle=3,color="Niagara
  Burgundy"):
  Graphe:=plot([x,g(x),x=-3..0]):
  plots[display]({DroiteV,Graphe},view=[-3..3,-1..100]);
```



Dans le cas d'une discontinuité infinie où le comportement des images des voisins de gauche et de droite d'une valeur de x n'est pas le même, on ne peut pas utiliser la notation limite sans préciser la "*l'approche*" des valeurs de x pour donner l'information quant au comportement des images de tous les voisins de x.

Examinons la fonction suivante:

```
> f:=x->1/(x-1);
```

$$f := x \mapsto \frac{1}{x-1} \quad \textbf{(4.2.4)}$$

```
> plot([x,f(x),x=-1..3],discont=true,view=[-1..3,-20..20]);
```



```
> Limit('f'(x),x = 1,left)=limit(f(x),x = 1,left);
```

$$\lim_{x \to 1-} f(x) = -\infty \quad \textbf{(4.2.5)}$$

```
> Limit('f'(x),x = 1,right)=limit(f(x),x = 1,right);
```

$$\lim_{x \to 1+} f(x) = \infty \quad \textbf{(4.2.6)}$$

Essayons de voir ce que répondra Maple lorsqu'on lui demandera d'évaluer la limite sans préciser "l'approche" des voisins de 1. Pouvez-vous le deviner?

```
> Limit('f'(x),x = 1)=limit(f(x),x = 1);
```

$$\lim_{x \to 1} f(x) = \mathit{undefined} \quad \textbf{(4.2.7)}$$

Maple a répondu "undefined" mais nous savions déjà que limite n'existe pas car la limite à gauche n'existe pas. Une autre bonne façon de répondre aurait été de dire que la limite n'existe pas car la limite à droite n'existe pas.

Discontinuité par oscillations

Terminons cette présentation en observant la fonction $y = \sin\left(\frac{1}{x}\right)$.

```
> f:=x->sin(1/x):
  plot([x,f(x),x=-Pi..Pi],view=[-3..3,-2..2]);
```



Bien sûr, $f(0)$ n'étant pas définie, la première condition de la continuité n'est pas satisfaite. Donc cette fonction n'est pas continue en $x = 0$. Mais analysons le type de discontinuité de cette fonction en $x = 0$. Traçons, de nouveau, le graphique de f en limitant davantage l'intervalle des valeurs de x.

```
> plot([x,f(x),x=-0.5..0.5],view=[-0.5..0.5,-1.5..1.5]);
```



```
> plot([x,f(x),x=-0.01..0.01],view=[-0.01..0.01,-1.5..1.5]);
```



Il semble que les images des voisins de $x = 0$ ne deviennent ni de plus en plus voisines d'un nombre L, ni de plus en plus grandes. On remarque plutôt que ces images oscillent continuellement entre les droites $y = -1$ et $y = 1$. Interrogeons Maple pour voir. Intéressons-nous d'abord aux voisins de gauche de 0.

```
> Limit('f'(x),x=0,left)=limit(f(x),x=0,left);
```

(4.3.1)

$$\lim_{x \to 0-} f(x) = -1..1 \quad \textbf{(4.3.1)}$$

Maple répond, à sa façon, que la limite demandée *n'existe pas*. De plus, il nous informe que les images de ces valeurs de x voisines mais inférieures à 0 oscillent entre $y = -1$ et $y = 1$. La réponse donnée est la notation Maple pour un objet de type intervalle fermé.

De la même manière, pour les voisins de droite,

```
> Limit('f'(x),x=0,right)=limit(f(x),x=0,right);
```

$$\lim_{x \to 0+} f(x) = -1..1 \quad \textbf{(4.3.2)}$$

En résumé:

```
> Limit('f'(x),x=0)=limit(f(x),x=0);
```

$$\lim_{x \to 0} f(x) = -1..1 \quad \textbf{(4.3.3)}$$

En bref, le comportement des images des valeurs de x près de 0 est le même aussi bien que avec les voisins de gauche que pour ceux de droite de 0.

Même après quelques tracés de f en voulant zoomer vers l'avant, on ne voit pas grand chose sur le graphique précédent au voisinage de 0. Voici une manière de mieux voir le comportement "erratique" des images en zoomant vers l'avant le graphique de f au voisinage de 0.

On génère d'abord une suite de valeurs de x voisines de 0. On contrôle alors les antécédents de la fonction f pour son tracé au voisinage de 0.

```
> Points:= evalf[40]([seq]([-0.01 + i/200*0.04,f(-0.01 + i/200*0.04)
  ],i=0..200)):
  plots[pointplot](Points,style=line,connect=true,view=[-0.01..0.01,
  -1.5..1.5],color="Niagara  Burgundy");
```



Le mot de la fin

Presque toutes les fonctions de notre niveau d'enseignement sont continues sur ***tout intervalle contenu dans leur domaine de définition*** :

– les fonctions polynomiales sont continues sur $\mathbb{R}$;

– les fonctions rationnelles sont continues sur tout intervalle contenu dans leur domaine de définition ;

– les fonctions exponentielles sont continues sur $\mathbb{R}$;

– les fonctions logarithme népérien et logarithme décimal sont continues sur $]0, +\infty[$;

– les fonctions racines n-èmes sont continues sur [0, +∞[;

– les fonctions sinus et cosinus sont continues sur $\mathbb{R}$, la fonction tangente est continue sur tout intervalle ne contenant pas un nombre de la forme $\frac{\pi}{2} + k\pi, k \in \mathbb{Z}$.

– la fonction valeur absolue est continue sur $\mathbb{R}$;

– ***toutes les fonctions obtenues par opérations (somme, produit, quotient) ou composition à partir de ces fonctions de référence sont aussi continues sur leur domaine de définition.***

Portons maintenant notre attention sur la fonction f définie par $f(x) = \sqrt{x^2 + x - 6}$.

Obtenons le domaine de la fonction f.

Pour que la racine carrée puisse être définie, il faut que le radicand représente un nombre positif. Alors, résolvons l'inéquation $x^2 + x - 6 \geq 0$. (La relation $\geq$ impose une résolution dans $\mathbb{R}$)

```
> solve(x^2+x-6>=0,x);
```

$$(-\infty, -3], [2, \infty) \qquad \textbf{(5.1)}$$

La fonction f est continue sur son domaine donc la fonction f est continue sur $]-\infty, -3] \cup [2, \infty[$. Évidemment, la continuité en $x = -3$ est une continuité à gauche et la continuité en $x = 2$ est une continuité à droite.

```
> f:=x->:-sqrt(x^2+x-6);
```

$$f := x \mapsto \sqrt{x^2 + x - 6} \qquad \textbf{(5.2)}$$

```
> Limit(f(x),x=-3,left)=limit(f(x),x=-3,left);
  Limit(f(x),x=2,right)=limit(f(x),x=2,right);
```

$$\lim_{x \to (-3)-} \sqrt{x^2 + x - 6} = 0$$

$$\lim_{x \to 2+} \sqrt{x^2 + x - 6} = 0 \qquad \textbf{(5.3)}$$

```
> plot([x,f(x),x=-6..5],numpoints=500,xtickmarks=10,view=[-6..5,0..5]);
```



Observons le résultat, ci-dessous, à la requête suivante: $\lim_{x \to (-3)+} f(x)$.

```
> Limit(f(x),x=-3,right)=limit(f(x),x=-3,right);
```

$$\lim_{x \to (-3)^+} \sqrt{x^2 + x - 6} = 0 \qquad \textbf{(5.4)}$$

À l'aide de LimitTutor de la sous-bibliothèque Student[Calculus1], voyons comment comment l'évaluateur a opéré ce calcul. À l'exécution de `LimitTutor` apparaîtra une boîte de dialogue. Cliquez sur le bouton `Toutes les étapes` puis ensuite, sur le bouton `Fermer`. À la fermeture de la boîte de dialogue, tout le développement sera transposé dans le document appelant.

```
> LimitTutor((f(x),x=-3,right));
```

$$\lim_{x \to -3} \sqrt{x^2 + x - 6}$$

$$
\begin{aligned}
&= \sqrt{\lim_{x \to -3} (x^2 + x - 6)} && [power] \\
&= \sqrt{\left(\lim_{x \to -3} x^2\right) + \left(\lim_{x \to -3} x\right) + \lim_{x \to -3} (-6)} && [sum] \\
&= \sqrt{\left(\lim_{x \to -3} x^2\right) + \left(\lim_{x \to -3} x\right) - 6} && [constant] \\
&= \sqrt{\left(\lim_{x \to -3} x^2\right) - 9} && [identity] \\
&= \sqrt{\left(\lim_{x \to -3} x\right)^2 - 9} && [power] \\
&= 0 && [identity]
\end{aligned}
$$

$$\lim_{x \to -3} \sqrt{x^2 + x - 6} = 0 \qquad \textbf{(5.5)}$$

Le domaine de simplification de Maple est l'ensemble des nombres complex mais, selon la documentation, la sous-bibliothèque Student[Calculus1] devrait réaliser les simplifications dans l'ensemble des nombres réels. Contrairement à la documentation, il semble que Maple ait réalisé tout de même un passage à la limite dans $\mathbb{C}$. Ce résultat n'est pas celui attendu dans $\mathbb{R}$. Le passage à la limite qu'a opéré l'évaluateur dans le calcul précédent n'est pas acceptable dans $\mathbb{R}$. En effet, si $x > -3$, $x^2 - 9 < 0$ et donc, $\sqrt{x^2 - 9} \notin \mathbb{R}$.

```
> Limit(f(x),x=-3,right)=limit(f(x),x=-3,right);
  Limit(f(x),x=2,left)=limit(f(x),x=2,left);
```

$$\lim_{x \to (-3)^+} \sqrt{x^2 + x - 6} = 0$$

$$\lim_{x \to 2^-} \sqrt{x^2 + x - 6} = 0 \qquad \textbf{(5.6)}$$

Avec l'environnement `RealDomain`, on a ceci:

```
> use RealDomain in Limit(f(x),x=-3,right)=limit(f(x),x=-3,right) end
  use;
  use RealDomain in Limit(f(x),x=-3,left)=limit(f(x),x=-3,left) end
  use;
```

$$\lim_{x \to (-3)^+} \sqrt{x^2 + x - 6} = undefined$$

(5.7)

$$\lim_{x \to (-3)^-} \sqrt{x^2 + x - 6} = \textit{undefined} \tag{5.7}$$

Mais, ceci aussi.

```
> use RealDomain in Limit(f(x),x=2,left)=limit(f(x),x=2,left) end use;
  use RealDomain in Limit(f(x),x=2,right)=limit(f(x),x=2,right) end
  use;
```

$$\lim_{x \to 2^-} \sqrt{x^2 + x - 6} = \textit{undefined}$$

$$\lim_{x \to 2^+} \sqrt{x^2 + x - 6} = 0 \tag{5.8}$$

La leçon qu'on doit retenir c'est qu'**il est primordial de considérer le domaine** d'une fonction avant d'en faire l'analyse. La macro-commande `discont` a pour résultat non pas des discontinuités avérées mais des candidats à la discontinuité.

```
> discont(f(x),x);
```

$$\{-3, 2\} \tag{5.9}$$

Or, la fonction f est discontinue sur $]-3, 2[$.